Лекция 7.

Динамическое программирование

Основные определения

Многие олимпиадные задачи, а также задачи практического программирования, являются задачами на перебор вариантов и выбор среди этих вариантов допустимого или наилучшего по тому или иному критерию. Однако рассмотреть все варианты, в силу чрезвычайно большого их количества, зачастую не представляется возможным [1].

К счастью, для ряда задач, сходных по формулировке с проблемами, действительно требующими полного перебора вариантов, можно найти гораздо более эффективное решение. Чаще всего в таких случаях решение сводится к нахождению решений подзадач меньшей размерности, которые запоминаются в таблице и никогда более не пересчитываются, а подзадачи большей размерности используют эти уже найденные решения. Такой метод называется динамическим программированием, еще его называют табличным методом. В общей же форме под динамическим программированием понимают процесс пошагового решения задачи оптимизации, при котором на каждом шаге из множества допустимых решений выбирается одно, которое оптимизирует заданную целевую или критериальную функцию. Иногда вместо оптимизационной тем же методом решается задача подсчета количества допустимых решений. В этом случае на каждом шаге вместо выбора оптимального решения производится суммирование решений подзадач меньшей размерности, причем они по формулировке не всегда полностью совпадают с исходной задачей (соответствующие примеры будут рассмотрены ниже). В обоих случаях найденное на текущем шаге решение обычно заносится в таблицу. Как правило, связь задач и подзадач формулируется в виде некоторого “принципа оптимальности” и выражается системой уравнений (рекуррентных соотношений).

Основы теории динамического программирования были заложены Р. Беллманом (см., например, [2]). Описание данного метода и примеры различных задач, которые можно им решать приведены также в [3-6]. Заметим, что слово программирование в приведенном названии (dynamic programming), также как и в “линейном программировании” (linear programming) не означает составление программ для компьютера.

Для решения задачи оптимизации, в которой требуется построить решение с максимальным или минимальным (оптимальным) значением некоторого параметра, алгоритм, основанный на динамическом программировании, можно сформулировать так:

1) выделить и описать подзадачи, через решение которых будет выражаться искомое решение,

2) выписать рекуррентные соотношения (уравнения), связывающие оптимальные значения параметра для подзадач,

3) вычислить оптимальное значение параметра для всех подзадач,

4) построить само оптимальное решение, используя полученную информацию.

Если нас интересует только значение параметра, то шаг 4 в алгоритме не нужен (такая ситуация характерна, например, для задач подсчета количеств допустимых вариантов или некоторых конфигураций, в том числе и комбинаторных). Однако, в случае необходимости построения самогó оптимального решения иногда приходится в процессе выполнения шага 3 алгоритма получать и хранить дополнительную информацию. Зачастую именно шаг 4 оказывается самым сложным при реализации подобных алгоритмов.

Условия применимости динамического программирования

Попробуем формализовать свойство задачи, которое позволяет применять указанный метод. Пусть мы нашли решение исходной задачи, тогда любая часть этого решения или любая подзадача, через которую выражается решение исходной задачи, также в свою очередь является решением аналогичной задачи меньшей размерности. То есть оптимальное решение задачи содержит оптимальные решения ее подзадач. Чтобы убедиться, что задача обладает этим свойством, надо показать, что, улучшая решение подзадачи, мы улучшим и решение исходной задачи.

Как только свойство оптимальности для подзадач установлено, обычно становится ясно, с каким именно множеством подзадач будет иметь дело алгоритм. А, так как динамическое программирование в сущности вычисляет решение для всех подзадач, то количество различных подзадач, непосредственно через которые выражается решение задач большей размерности, должно быть таким, чтобы было возможно все их (или по крайней мере ту часть, которая непосредственно необходима для решения задач большей размерности) запомнить в таблице. Это и есть второе свойство задачи, существенное при использовании динамического программирования. Благодаря этому свойству при рекурсивном способе решения подобных проблем мы многократно выходим на одни и те же подзадачи. В таком случае говорят, что у переборной задачи (оптимизационной или подсчета количеств) имеются перекрывающиеся подзадачи. В типичных случаях (но не всегда !!!) количество различных подзадач полиномиально зависит от размера исходных данных. Алгоритм, основанный на динамическом программировании, решает каждую из подзадач единожды и заносит его в таблицу. Когда эта же подзадача встречается снова, программа не тратит время на ее решение, а берет готовый ответ из таблицы. Поэтому общее время решения пропорционально размеру заполняемой таблицы в целом, хотя зачастую на этапе реализации алгоритма хранению подлежит лишь ее часть, например, одна строка. В любом случае это время существенно меньше времени перебора всех возможных вариантов с целью выбора среди них лучшего или подсчета общего количества допустимых вариантов.

Классические задачи динамического программирования

Задача 1. В таблице NхN, клетки заполнены случайным образом цифрами от 0 до 9. Найти маршрут из клетки A(1,1) в клетку A(N, N) такой, что:

1) он будет состоять из отрезков, соединяющих центры клеток, имеющих общую сторону

2) длина маршрута минимально возможная

3) из всех маршрутов, удовлетворяющих условиям (1) и (2), искомый маршрут тот, сумма цифр в клетках которого максимальна.

Решение. Пусть клетка (1, 1) это левый верхний угол таблицы, а (N, N), соответственно, правый нижний угол. Из условия (2) задачи следует, что за каждый шаг мы будем продвигаться по таблице либо на шаг вправо, либо на шаг вниз, что сразу нам гарантирует минимальность в длине пути и избавляет от анализа вариантов по данному критерию. Рассмотрим произвольную клетку таблицы (i, j). В нее мы можем прийти или из клетки (i-1, j), или из (i, j-1). Тогда, если мы уже знаем оптимальные маршруты из клетки (1, 1) в каждую из этих двух клеток, то оптимальным маршрутом в клетку (i, j) будет подмаршрут с максимальной из двух сумм суммой плюс отрезок, соединяющий (i, j) с концом выбранного подмаршрута. Оптимальные маршруты из (1, 1) в (1, 2) и (2, 1) определены однозначно. Зная их, по указанному выше способу мы найдем оптимальные маршруты в (1, 3), (2, 2), (3, 1) и запишем их в соответствующих клетках таблицы (записывать нужно только сумму цифр маршрута и направление его последнего отрезка). Этот процесс можно продолжить пока вся таблица не будет заполнена, причем заполнять ее можно по строчкам слева направо. В клетке (N, N) мы в итоге получим значение суммы цифр искомого маршрута и последний его отрезок. По такой таблице легко восстановить и весь маршрут, начиная с клетки (N, N). Рассмотрим любую часть оптимального маршрута, например, между клетками (i1, j1) и (i2, j2). Докажем, что эта часть маршрута является решением исходной задачи для указанных клеток. Пусть это не так и существует маршрут с большей суммой, соединяющий эти клетки и имеющий такую же длину. Тогда и для клеток (1, 1) и (N, N) мы можем построить лучший маршрут, используя отрезки, соединяющие (1, 1) и (i1, j1), а также (i2, j2) и (N, N) из старого маршрута плюс улучшенный маршрут из (i1, j1) в (i2, j2), а это противоречит тому, что мы изначально рассматривали часть из уже оптимального маршрута. Значит, любая часть оптимального маршрута в свою очередь является оптимальной.

Программа решения данной задачи приведена, например, в [1].

Задача 2. Наибольшая общая подпоследовательность (НОП).

Подпоследовательность можно получить из некоторой конечной последовательности, если удалить из последней некоторое множество ее элементов (возможно пустое). Например, BCDB является подпоследовательностью последовательности ABCBDAB. Будем говорить, что последовательность Z является общей подпоследовательностью последовательностей X  и Y, если Z является подпоследовательностью как X,  так и Y. Требуется для двух последовательностей X  и Y найти общую подпоследовательность наибольшей длины. Заметим, что НОП может быть несколько.

Решение. Задача о НОП обладает свойством оптимальности для подзадач. Здесь подходящее множество подзадач — множество пар префиксов (начальных частей) двух данных последовательностей. Покажем это. Пусть Z = {z1, z2, …, zk} одна из НОП для X = {x1, x2, …, xm} и Y = {y1, y2, …, yn}. Если xm = yn, то zk = xm = yn (в противном случае мы могли бы дописать xm = yn в конец последовательности Z и получить НОП длины k+1. Кроме того Zk-1 является НОП для Xm-1 и Yn-1 (если это не так, то мы можем к НОП Xm-1 и Yn-1 дописать xm = yn и получить НОП для X и Y более длинную, чем Z. Если zk ( xm, то так как Z — НОП для X и Y, она тем более является НОП для Xm-1 и Y. Аналогично, если zk ( yn, то Z — НОП для X и Yn-1. Максимальное количество подзадач, которое нам может понадобиться решить, равно m(n (для каждой пары префиксов X и Y). Это позволяет использовать при решении динамическое программирование. Выпишем рекуррентное соотношение между длинами НОП в подзадачах, обозначив за a[i, j] длину НОП для Xi и Yj:

Приведем программу решения этой задачи для последовательностей, состоящих из не более чем 250 символов:

var x,y,z:string;

    a:array[0..250,0..250] of byte;

    i,j:byte;

begin

  readln(x);

  readln(y);

  fillchar(a,sizeof(a),0);

  for i:=1 to length(x) do

  for j:=1 to length(y) do

    if x[i]=y[j] then a[i,j]:=a[i-1,j-1]+1 

    else if a[i-1,j]>=a[i,j-1] then a[i,j]:=a[i-1,j]

                               else a[i,j]:=a[i,j-1];

  {длина НОП найдена в a[length(x),length(y)]}

  z:='';{строим саму НОП}

  i:=length(x);

  j:=length(y);

  while (i>0) and (j>0) do

  if x[i]=y[j] then begin z:=x[i]+z; i:=i-1; j:=j-1 end

  else if a[i-1,j]>=a[i,j-1] then i:=i-1 else j:=j-1;

  writeln(z)

end.
Задача 3. Задача об оптимальной расстановке скобок.

Рассмотрим ее, например, в такой формулировке. В арифметическом выражении, операндами которого являются целые числа, а операциями — бинарные арифметические операции “+” и “(”, расставить скобки так, чтобы результат оказался максимальным (исключение других арифметических операций не снижает в данном случае общности рассмотрения проблемы).

Решение. Прежде чем применять динамическое программирование к этой задаче, стоит убедиться, что простой перебор всех расстановок скобок не даст эффективного алгоритма. Из комбинаторики известно, что количество различных расстановок скобок между n операндами Pn равно числу Каталана с номером n – 1 (см., например, [7]), а именно:

где с — некоторая константа, не зависящая от n (такую нижнюю оценку обозначают ((4n/n3/2) [4]). То есть, число расстановок скобок экспоненциально зависит от n, так что полный перебор неэффективен.

Подзадачами, через которые мы будем выражать оптимальное решение, будут задачи об оптимальной расстановке скобок в произведениях наших операндов, начиная с i-го и заканчивая j-м. Запоминать результат оптимального решения соответствующей подзадачи мы будем в элементе a[i, j] матрицы, размером n(n, диагональные элементы которой (a[i, i]) равны операндам, а для i > j все элементы равны 0. Пусть мы хотим подсчитать значение арифметического выражения для операндов, начиная с i-го и заканчивая j-м для i < j. Если мы предположим, что последней будет выполняться арифметическая операция, расположенная после операнда с номером k (i ( k < j) и эта операция — сложение, то результат подсчета будет равен сумме элементов a[i, k] и a[k+1, j] нашей матрицы. С умножением ситуация несколько более сложная. Так как операндами могут быть и отрицательные числа, то для нахождения максимума из произведений нужно знать не только максимальные, но и минимальные значения для арифметических операций над операндами с i-го по k-й и с k+1-го по j-й соответственно. Значит для любой последовательности операндов нужно хранить значение как максимально, так и минимально возможного результата выполнения арифметических операций над ними. Однако, дополнительная матрица для хранения минимальных значений не нужна. Так как для i > j матрица a не заполнена, мы можем отвести эти элементы для хранения минимумов, только запоминать результат решения этой подзадачи для операндов, начиная с i-го и заканчивая j-м, мы будем в элементе a[j, i]. Тогда в общем случае, если после k-го операнда стоит операция умножения (i ( k < j), то максимальный результат будет равен max(a[i, k]( a[k+1, j], a[k, i]( a[j, k+1], a[i, k]( a[j, k+1], a[k, i]( a[k+1, j]). Минимальный результат вычисляется аналогично. Оптимальное значение k нам не известно, но число k может принимать всего i – j различных значений. Поскольку одно из них оптимально, то следует перебрать их все и выбрать наилучшее. Таким образом, любой элемент таблицы вычисляется за не более, чем 4|i – j| операций, что соответствует асимптотической оценке O(n). Единственным неудобством рассмотренного подхода является необходимость заполнять матрицу по линиям, параллельным главной диагонали: сначала вычисляются элементы, для которых |i – j| = 1, потом — |i – j| = 2 и т. д. Искомое значение арифметического выражения будет получено в элементе a[1, n], который заполняется в последнюю очередь. Общее время работы алгоритма есть O(n3), тем самым он значительно эффективнее перебора всех возможных порядков выполнения арифметических операций. Объем памяти, необходимый для хранения таблицы, есть O(n2).
Приведем фрагмент программы для решения этой задачи, в котором и заполняется матрица a. Для хранения значений арифметических операций после каждого операнда используется одномерный массив op:

for m:=1 to n-1 do {m=|i-j|}

  for i:=1 to n-m do

    begin

      j:=i+m;

      r_max:=-maxlongint-1;

      r_min:=maxlongint;

      for k:=i to j-1 do

        case op[k] of

        '+':begin {используем определенные выше

            функции нахождения максимума и минимума

            из двух элементов}

              r_max:=max(r_max,a[i,k]+a[k+1,j]);

              r_min:=min(r_min,a[k,i]+a[j,k+1])

            end;

        '*':begin

              r_max:=max(r_max,

                     max(a[i,k]*a[k+1,j],

                     max(a[k,i]*a[j,k+1],

                     max(a[i,k]*a[j,k+1],

                         a[k,i]*a[k+1,j]))));

              r_min:=min(r_min,

                     min(a[i,k]*a[k+1,j],

                     min(a[k,i]*a[j,k+1],

                     min(a[i,k]*a[j,k+1],

                         a[k,i]*a[k+1,j]))))

            end;

        end;{case}

      a[i,j]:=r_max;

      a[j,i]:=r_min

    end;

  writeln(a[1,n]);
Если помимо нахождения максимально возможного значения арифметического выражения требуется напечатать и сам оптимальный вариант расстановки скобок, то потребуется хранить еще и значения k, на которых достигается максимум или минимум. Попробуйте самостоятельно написать рекурсивный алгоритм, печатающий результат расстановки скобок требуемым образом.

Задача 4. Оптимальная триангуляция выпуклого многоугольника.

Требуется разрезать выпуклый n-угольник n – 2 непересекающимися диагоналями на треугольники так, чтобы сумма длин диагоналей (стоимость разрезания) была минимальной.

Решение. Несмотря на геометрический характер, эта задача легко сводится к предыдущей. Если мы проведем в таком многоугольнике одну диагональ, то стоимость разрезания, в котором эта диагональ обязательно присутствует, можно выразить через стоимость разрезания образовавшихся при ее проведении двух многоугольников. Более подробно о решении этой задачи можно прочитать в [3,4,6].

Волновые алгоритмы

Обычно под волновыми алгоритмами понимают некоторые способы поиска кратчайшего пути в графе. Рассмотрим их с точки зрения динамического программирования (ведь они относятся именно к этому методу решения задач), обобщив на более широкий класс проблем. Заметим, что при реализации волновых алгоритмов зачастую часть таблицы, отведенная для запоминания решений подзадач, остается незаполненной, а часть действий алгоритма является лишней, что отличает их от решений задач, рассмотренных выше. Однако именно возможность использовать избыточную память и увеличить количество производимых операций, например в n раз, делает программы, реализующие эти алгоритмы чрезвычайно простыми. Давайте в этом убедимся.

Задача 5. Робот. (Командная олимпиада г. Санкт-Петербурга по программированию, 1999 г.)

В исследовательской лаборатории фирмы Robots&Co разработали новую модель робота. Он работает по заранее заданной программе, в которой могут присутствовать команды: сделать шаг на Юг, на Север, на Восток или на Запад. Робот исполняет программу строго последовательно и, дойдя до конца программы, останавливается. Специалисты из Robots&Co заинтересовались вопросом, сколько существует различных программ, состоящих из K инструкций, таких, что робот, выйдя из начала координат, придет в точку с координатами (X, Y) (

). Оси координат располагаются параллельно сторонам света, и единица измерения соответствует одному шагу робота. Напишите программу, которая дает ответ на этот вопрос.

Решение. Решать задачу будем волновым алгоритмом. А именно, сначала посчитаем сколько существует программ, состоящих из одной инструкции, причем заканчивающихся во всех возможных точках (вернее в тех, до которых успеет “докатиться волна” из начала координат за один шаг). Затем, опираясь на результаты первого шага, рассмотрим все программы, состоящие из двух инструкций и т.д. Результаты будем запоминать в матрице a. Тогда, если после L шагов в каждом элементе a[i, j] матрицы (-16 ( i, j ( 16) будет храниться количество программ, состоящих из L инструкций и заканчивающихся в точке с координатами (i, j), то для каждой из точек (i, j) легко подсчитать количество программ, заканчивающихся в ней и состоящих из L+1 инструкции, по формуле a[i+1, j] + a[i, j+1] + a[i-1, j] + a[i, j-1]. Теперь несложно написать программу для решения этой задачи. Для того, чтобы избежать проверок на выход за границу массива, как обычно используем барьерный метод (увеличиваем размеры таблицы на единицу во всех направлениях):

var x,y,k:integer;

  a,b:array[-17..17,-17..17] of longint;

  i,p,q:integer;

begin

  read(k,x,y);

  fillchar(a,sizeof(a),0);

  b:=a;

  a[0,0]:=1;

  for i:=1 to k do

    begin

      for p:=-i to i do {остальные точки заведомо недостижимы}

        for q:=-i to i do

          b[p,q]:=a[p+1,q]+a[p,q+1]+a[p-1,q]+a[p,q-1];

      a:=b

    end;

  writeln(a[x,y])

end.

Количество операций, которые выполняет данный алгоритм есть O(Kn2), где n — размер просматриваемой на каждом шаге таблицы. Если, как в нашем случае, K не превосходит n, то количество операций растет как O(n3), хотя для данной задачи возможно построить алгоритм с вычислительной сложностью O(n2). Такой алгоритм для решения подобных задач будет рассмотрен в одной из следующих лекций. Вообще говоря, с точки зрения предложенного решения, K может быть почти как угодно большим. Для динамической схемы в данном случае важно лишь то, что робот может перемещаться в ограниченной области, для описания которой мы в состоянии завести соответствующую таблицу. Хотя часть ее так и останется незаполненной (в нашей таблице всегда найдутся клетки, до которых робот по условию дойти за K шагов не сможет), но в просмотрах на каждом шаге алгоритма такие клетки участвуют, несмотря на то что мы попытались ограничить для каждого шага просматриваемую часть таблицы. Дальнейшая же оптимизация алгоритма приведет к потере прозрачности решения.

Задача 6. Лабиринт.

Класс задач о лабиринтах достаточно широк и возможны различные формулировки подобных задач. Однако их решения мало отличаются друг от друга. Поэтому рассмотрим задачу о лабиринте в следующей формулировке. (Московская олимпиада по программированию, 1983 г.)

Лабиринт задан массивом a размером n(n, в котором a[i, j]=1, если клетка “проходима”, и a[i, j]=0 — если “непроходима”. Путник изначально размещается в “проходимой” клетке [i0, j0]. Он может перемещаться в любую из соседних “проходимых клеток”, если у нее есть общая сторона с той, в которой он находится. Определить, может ли путник выйти из лабиринта. Если может, то напечатать путь от выхода до начального положения путника. Выходом считается любая граничная точка массива a.

Решение. Запишем в клетку [i0, j0] число 2. Просмотрим все клетки лабиринта (или исключим из рассмотрения заведомо недостижимые за один ход). Если у рассматриваемой “проходимой” клетки, помеченной единицей есть сосед, помеченный двойкой (в общем случае числом k), то мы запишем в нее 3 (в общем случае — k+1). Таким образом на каждом шаге алгоритма числом k будут помечены все те клетки, до которых путник может добраться ровно за k-2 единичных перемещения (до которых за k-2 шага “докатилась волна”). Этот процесс закончится, когда либо очередное число будет вписано в граничную клетку, либо на за весь просмотр массива ни одна из клеток не будет помечена (выхода в этом случае нет). Печать решения также произвести несложно. Если последняя клетка помечена числом k, то предпоследней в пути может быть любая, соседняя с ней клетка, помеченная числом k-1 и т.д. Полученный путь по построению является кратчайшим. Заметим, что для решения этой задачи дополнительные массивы можно не заводить, однако просмотр ряда клеток массива, как и в предыдущей задаче будет избыточным. Приведем основную часть программы для решения этой задачи, в которой как и ранее используется нулевой “барьер”:

k=2; a[i0,j0]:=k;

flag:=true;{были помечены новые клетки}

while flag do

begin

  flag:=false;

  for i:=i0-(k-1) to i0+k-1 do

    for j:=j0-(k-1) to j0+k-1 do

      if (a[i,j]=1) and ((a[i-1,j]=k)or

         (a[i+1,j]=k)or(a[i,j-1]=k)or

         (a[i,j+1]=k)) then 

      begin 
        a[i,j]:=k+1; flag:=true;

        if (i in [1,n])or(j in [1,n]) then goto 1

      end;

  k:=k+1
end;

1:if flag then {выход найден}

    repeat {печатаем путь}

      writeln(i,’ ’,j);

      if a[i-1,j]=k then i:=i-1 else

      if a[i,j-1]=k then j:=j-1 else

      if a[i+1,j]=k then i:=i+1 else j:=j+1;

      k:=k-1

    until k=0;{[i0,j0] напечатана}

В данной программе использование оператора goto иллюстрирует единственный случай его допустимого применения в структурированной программе. А именно: необходимость досрочного выхода сразу из нескольких вложенных циклов (в нашем примере — трех).

Для того чтобы исключить лишние просмотры клеток массива, можно завести список (динамический или организованный с помощью одномерного массива) и хранить в нем координаты только тех клеток, рассмотрение которых следует произвести на очередном шаге.
Задача 7. В некоторой игре одно двузначное число можно заменить на другое по следующему правилу: любая из двух цифр исходного числа заменяется на сумму или разность его цифр (в случае разности из большей цифры вычитается меньшая). Для двузначных чисел а и b построить последовательность чисел минимальной длины, начинающуюся с числа a, заканчивающуюся b, а каждое следующее число в цепочке можно получить из предыдущего по указанному выше правилу или указать, что это сделать невозможно. Например, для чисел 12 и 31 последовательность будет выглядеть так: 12 32 31.

Решение. Отнести решение этой задачи к методу динамического программирования позволяет следующее свойство. Если допустимая последовательность минимальной длины построена, то любая ее часть решает аналогичную задачу для начального и конечного числа в выделенной подпоследовательности. Для реализации решения следует завести одномерный массив (таблицу), состоящий из 99 элементов, каждый элемент массива соответствует одному из двузначных чисел (здесь под двузначными понимаются все натуральные числа, не превосходящие 99). Первоначально массив следует обнулить, а в элемент массива с индексом a занести 1. Далее, получив из числа a все возможные вторые элементы последовательности, следует занести в соответствующие элементы массива двойки. Произвольный же (i-ый) шаг алгоритма будет выглядеть так. Просматриваем массив и, если какой-либо его элемент равен i-1, то из индекса этого элемента получаем все допустимые варианты преобразованных чисел и записываем в соответствующие им элементы массива число i, но только если ранее значение элемента было равно 0. Алгоритм заканчивается, если будет заполнен элемент с индексом b, или на i-ом шаге ни один ранее нулевой элемент массива не будет заполнен числом i (решение в таком случае отсутствует). Если решение существует, то восстановить его по описанному массиву несложно. Динамическое программирование в данном случае удалось применить лишь потому, что количество различных чисел, которые могут возникнут по ходу игры ограничено (их всего 100). Размер таблицы поэтому очевиден, однако на каждом шаге бóльшая часть элементов, в данном случае одномерного массива, будет просматриваться зря. Изменить это можно так же, как и в предыдущей задаче.
Задача 8. Арифметическим показателем натурального числа M по цифре N (записывается Ar(M, N)) назовем минимальное количество цифр N, которые необходимо использовать в некотором арифметическом выражении, содержащем только цифры N, знаки операций +, - , * , / и скобки, дающем результат M. Например, Ar(17, 3)=5, а соответствующее выражение может быть, например, следующим: 3*(3 + 3) – 3/3, а Ar(24, 1)=5, поскольку существует выражение (11 + 1)*(1 + 1)=24. Вычислите арифметический показатель для заданной пары чисел. Результат любой промежуточной операции должен быть натуральным и меньше 1000. (Московская командная олимпиада по программированию, 1999г.)

Решение. Как и в предыдущей задаче, применение динамического программирования возможно благодаря ограничению не только на результат, но и на промежуточные операции. Более того, решение данной задачи вообще не отличалось бы от предыдущей (за исключением правила, по которому формируются клетки для заполнения на очередном шаге алгоритма, в данном случае k-й шаг соответствует k цифрам в арифметическом выражении), если бы цифры не разрешалось объединять в многозначные числа. При имеющихся ограничениях цифры могут образовывать лишь двузначные и трехзначные числа. Поэтому, для того чтобы получить все допустимые числа из k цифр, следует рассмотреть все элементы вспомогательного массива (размерностью 1000), помеченные числом k – 1, и по очереди произвести все арифметические операции между их индексами и заданной в условии цифрой (для вычитания и деления в ряде случаев следует рассмотреть оба варианта взаимного расположения операндов). Потом рассмотреть все элементы, помеченные числом k – 2, и произвести все арифметические операции между их индексами и двузначным числом, образованным при слиянии двух заданных цифр. Наконец, следует рассмотреть элементы, помеченные числом k – 3, и произвести арифметические операции между их индексами и трехзначным числом, образованным при слиянии трех заданных цифр. Если при этом в результате арифметической операции получается натуральное число, меньшее 1000, и соответствующий элемент вспомогательного массива равен 0, то он помечается числом k. Таким образом, данный алгоритм также относится к “волновым”.

Рассмотрение задач динамического программирования будет завершено в следующей лекции.

